Schwap CPU Design Documentation

Charlie Fenoglio, Alexander Hirschfeld, Andrew McKee, and Wesley Van Pelt

Winter 2015/2016

1 Registers

There are a total of 76 16-bit registers; 12 are fixed and 64 (spilt into 16 groups of 4) "schwapable" registers. Some registers have alias names, see Section 4.3.1 for a list.

1.1 Register Names and Descriptions

Name	Number	Description	Saved Across Call?
$z0	0	The Value 0†	-
$a0	1	Assembler Temporary 0	No
$a1	2	Assembler Temporary 1	No
$pc	3	Program Counter‡	Yes
$sp	4	Stack Pointer	Yes
$ra	5	Return Address	Yes
$s0 - $s1	6 - 7	User Saved Temporaries	Yes
$t0 - $t3	8 - 11	User Temporaries	No
$h0 - $h3	12 - 15	Schwap	-

†See Section 4.1.1-1 for details
‡See Section 4.1.1-2 for details

1.2 Schwap Registers

The "schwap" registers are registers that appear to be swapped using a command. There is no data movement when schwapping, it only changes which registers the $h0 - $h3 refer to. There are 8 groups the user can use for general purpose and 8 reserved groups.

1.2.1 Schwap Group Numbers, Descriptions, and Uses

Group Number	Uses	Saved Across Call?
0 - 3	User Temporaries	No
4 - 7	User Saved Temporaries	Yes
8	Arguments 0 - 3	No
9	Return Values 0 - 3	No
10 - 14	Reserved For Future Use†	-
15	Go to the last used group	-

†See Section 4.1.2-1 for details

2 Instructions

All instructions are 16-bits. The destination register is also used as a source unless otherwise noted. All offsets are bit shifted left by 1 since all instructions are 2 bytes long.

2.1 Instruction Types and Bit Layouts

Instructions can be manually translated by putting the bits for each of the components of the instructions in the places listed by the diagrams for each type. The OP codes, function codes, and types can by found on the "Core Instructions Summary" (2.2.1) table. The destination and source are register numbers, which can be found under the "Register Names and Descriptions" (1.1) table. Schwap group numbers can be found under the "Schwap Group Numbers, Descriptions, and Uses" (1.2.1) table. The active schwap group is not preserved over a function call. See Section 4.2.1 for notes on the types and layouts.

2.1.1 A-Type



Used for all ALU operations. It consists of a 4-bit OP code, 4-bit destination, 4-bit source, and a 4-bit function code. If the instruction has an immediate, it is inserted as the next instruction.

2.1.2 B-Type



If it is being used for branching it consists of a 4-bit OP code, 4-bit 1st source (R0), 4-bit 2nd source (R1), and a 4-bit (unsigned) offset. If it is being used for reading and writing to memory, it consists of a 4-bit OP code, 4-bit destination (R0 not used as a source), 4-bit source (R1), and a 4-bit (unsigned) offset.

2.1.3 H-Type

Used for schwapping. It consists of a 4-bit OP code, 8 unused bits, and a 4-bit schwap group number.

2.1.4 J-Type

Used for jumping. It consists of a 4-bit OP code, 4-bit source, and an 8-bit (signed) offset.



2.1.5 S-Type

Used for sudo. Only has a 4-bit OP code.



2.2 Core Instructions

Some instructions have alias names, see Section 4.3.2 for a list.

2.2.1 Core Instructions Summary

OP Code	Function Code	Name	Type	Description
0x0/1†	0x0	and	A	Bitwise ands 2 values
	0x1	orr	A	Bitwise ors 2 values
	0x2	xor	A	Bitwise xors 2 values
	0x3	not	A	Bitwise nots 2 values
	0x4	tsc	A	Converts a number to 2's compliment
	0x5	slt	A	Set less than
	0x6	sll	A	Left logical bit shift
	0x7	srl	A	Right logical bit shift
	0x8	sra	A	Right arithmetic bit shift
	0x9	add	A	Adds 2 values
	0xA	sub	A	Subtracts 2 values
	0xF	cpy	A	Copies the value in one register to another
0x2	-	beq	B	Branches if the 2 values are equal
0x3	-	bne	B	Branches if the 2 values are not equal
0x4	-	bgt	B	Branches if value0 > value1
0x5	-	blt	B	Branches if value0 < value1
0x6/7‡	-	j	J	Jumps to the value
0x8	-	r	B	Reads the value in memory into a register
0x9	-	w	B	Writes the value in a register into memory
0xE	-	rsh	H	Changes the schwap group
0xF	-	sudo	S	Sames as "syscall" in MIPS

†0x0 is used for instructions which do not use an immediate, 0x1 is used if the instruction does use an immediate
‡0x6 is used for instructions which jump to an address in a register, 0x7 is used if the instruction jumps to a label

2.2.2 Core Instructions and Details

and: A standard bitwise "and", it can use any of the syntaxes listed below.

Syntax	Meaning	Description
and [dest] [src]	dest = dest & src	Bitwise ands the values in registers [dest] and [src]
and [dest] [src] [immediate]	src = immediate dest = dest & src	Loads the immediate into the register [src] and then bitwise ands the values in registers [dest] and [src]
and [dest] [immediate]	dest = dest & immediate	Bitwise ands the immediate and the value in the register [dest]

orr: A standard bitwise "or", it can use any of the syntaxes listed below.

Syntax	Meaning	Description
orr [dest] [src]	dest = dest | src	Bitwise ors the values in registers [dest] and [src]
orr [dest] [src] [immediate]	src = immediate dest = dest | src	Loads the immediate into the register [src] and then bitwise ors the values in registers [dest] and [src]
orr [dest] [immediate]	dest = dest | immediate	Bitwise ors the immediate and the value in the register [dest]

xor: A standard bitwise "xor", it can use any of the syntaxes listed below.

Syntax	Meaning	Description
xor [dest] [src]	dest = dest ^ src	Bitwise xors the values in registers [dest] and [src]
xor [dest] [src] [immediate]	src = immediate dest = dest ^ src	Loads the immediate into the register [src] and then bitwise xors the values in registers [dest] and [src]
xor [dest] [immediate]	dest = dest ^ immediate	Bitwise xors the immediate and the value in the register [dest]

not: A standard bitwise "not", it can use any of the syntaxes listed below.

Syntax	Meaning	Description
not [dest]	dest = ∼dest	Bitwise nots the value in the register [dest]
not [dest] [src] [immediate]	src = immediate dest = ∼src	Loads the immediate into the register [src] and then bitwise nots the value in register [src] into [dest]
not [dest] [immediate]	dest = ∼immediate	Bitwise nots the immediate into the register [dest]

tsc: Converts an integer to/from 2's compliment, it can use any of the syntaxes listed below.

Syntax	Meaning	Description
tsc [dest]	dest = ∼dest + 1	Converts the value in the register [dest] to 2's compliment
tsc [dest] [src] [immediate]	src = immediate dest = ∼src + 1	Loads the immediate into the register [src] and then converts the value in register [src] to 2's compliment and stores into [dest]
tsc [dest] [immediate]	dest = ∼immediate + 1	Converts the immediate to 2's compliment and stores into the register [dest]

slt: A standard ALU set less than, it can use any of the syntaxes listed below.

Syntax	Meaning	Description
slt [dest] [src]	dest = (dest < src) ? 1 : 0	If [dest] < [src], then [dest] gets set to 1 If [dest] ≥ [src], then [dest] gets set to 0
slt [dest] [src] [immediate]	src = immediate dest = (dest < src) ? 1 : 0	Loads the immediate into the register [src] then If [dest] < [src], then [dest] gets set to 1 If [dest] ≥ [src], then [dest] gets set to 0
slt [dest] [immediate]	dest = (dest < immediate) ? 1 : 0	If [dest] < [immediate], then [dest] gets set to 1 If [dest] ≥ [immediate], then [dest] gets set to 0

sll: A standard shift left logical, it can use any of the syntaxes listed below.

Syntax	Meaning	Description
sll [dest] [shift]	dest = dest << shift	Shifts the value in the register [dest] left logically by [shift]
sll [dest] [shift] [immediate]	shift = immediate dest << shift	Loads the immediate into the register [src] and then shifts the value in the register [dest] left logically by [shift]
sll [dest] [immediate]	dest = dest << immediate	Shifts the value in the register [dest] logically by [immediate]

srl: A standard shift right logical, it can use any of the syntaxes listed below.

Syntax	Meaning	Description
srl [dest] [shift]	dest = dest >> shift	Shifts the value in the register [dest] right logically by [shift]
srl [dest] [shift] [immediate]	shift = immediate dest >> shift	Loads the immediate into the register [src] and then shifts the value in the register [dest] right logically by [shift]
srl [dest] [immediate]	dest = dest >> immediate	Shifts the value in the register [dest] right logically by [immediate]

sra: A standard shift right arithmetic, it can use any of the syntaxes listed below.

Syntax	Meaning	Description
srl [dest] [shift]	dest = dest >>> shift	Shifts the value in the register [dest] right arithmetically by [shift]
srl [dest] [shift] [immediate]	shift = immediate dest >>> shift	Loads the immediate into the register [src] and then shifts the value in the register [dest] right arithmetically by [shift]
srl [dest] [immediate]	dest = dest >>> immediate	Shifts the value in the register [dest] right arithmetically by [immediate]

add: A standard ALU add, it can use any of the syntaxes listed below.

Syntax	Meaning	Description
add [dest] [src]	dest = dest + src	Adds the values in registers [dest] and [src]
add [dest] [src] [immediate]	src = immediate dest = dest + src	Loads the immediate into the register [src] and then adds the values in registers [dest] and [src]
add [dest] [immediate]	dest = dest + immediate	Adds the immediate and the value in the register [dest]

sub: A standard ALU subtract, it can use any of the syntaxes listed below.

Syntax	Meaning	Description
sub [dest] [src]	dest = dest − src	Subtracts the values in registers [dest] and [src]
sub [dest] [src] [immediate]	src = immediate dest = dest − src	Loads the immediate into the register [src] and then subtracts the values in registers [dest] and [src]
sub [dest] [immediate]	dest = dest − immediate	Subtracts the immediate and the value in the register [dest]

cpy: Copies the value from one register to another, or loads an immediate. It can use any of the syntaxes listed below.

Syntax	Meaning	Description
cpy [dest] [src]	dest = src	Copies the value the in register [src] into [dest]
cpy [dest] [immediate]	dest = immediate	Loads the immediate into the register [dest]

beq: Branches if the 2 values are equal. PC relative, if branching up or down more than 16 instructions, this will be a pseudo instruction. It uses the syntax listed below.

Syntax	Meaning	Description
beq [src0] [src1] label	if(src0 == src1) goto label	If the values in the registers [src0] and [src1] are equal, branch to label

bne: Branches if the 2 values are not equal. PC relative, if branching up or down more than 16 instructions, this will be a pseudo instruction. It uses the syntax listed below.

Syntax	Meaning	Description
bne [src0] [src1] label	if(src0 != src1) goto label	If the values in the registers [src0] and [src1] are not equal, branch to label

bgt: Branches if the value0 > value1. PC relative, if branching up or down more than 16 instructions, this will be a pseudo instruction. It uses the syntax listed below.

Syntax	Meaning	Description
bgt [src0] [src1] label	if(src0 > src1) goto label	If [src0] > [src1], branch to label

blt: Branches if the value0 < value1. PC relative, if branching up or down more than 16 instructions, this will be a pseudo instruction. It uses the syntax listed below.

Syntax	Meaning	Description
blt [src0] [src1] label	if(src0 < src1) goto label	If [src0] < [src1], branch to label

j: Jumps to an instruction. It can use any of the syntaxes listed below.

Syntax	Meaning	Description
j label	-	Jumps to the instruction at label
j [dest]	pc = dest	Jumps to the instruction at the address in [dest]

r: Reads a value in memory, it uses the syntax listed below.

Syntax	Meaning	Description
r [dest] [src] [offset]	dest = Mem[src + offset]	Reads the data in the address of [src] + [offset] in memory into [dest]

w: Writes a value in memory, it uses the syntax listed below.

Syntax	Meaning	Description
w [dest] [src] [offset]	Mem[dest + offset] = src	Writes the data in the address of [dest] + [offset] in memory from [src]

rsh: Sets the schwap group, it uses the syntax listed below.

Syntax	Meaning	Description
rsh [group]	SchwapGroup = group	Changes the schwap group number to [group], these numbers can be found in the table in 1.2.1

sudo: Does what "syscall" in MIPS does, just type "sudo" for the instruction.

2.3 Pseudo Instructions

There are two types of pseudo instructions. One are instructions which are always pseudo instructions, the other are sometimes pseudo depending on the conditions. Some instructions have alias names, see Section 4.3.2 for a list.

2.3.1 Always Pseudo Instructions

Name	Syntax	Actual Code	Description
jal	jal label	cpy $ra, $pc j [label]	Stores the return address and then jumps to the label
bge	bge [src0] [src1] label	cpy $a0 [src0] slt $a0 [src1] beq $a0 $z0 label	If [src0] $\geq$ [src1], branch to label
ble	ble [src0] [src1] label	cpy $a0 [src1] slt $a0 [src0] beq $a0 $z0 label	If [src0] $\leq$ [src1], branch to label

2.3.2 Conditional Pseudo Instructions

Name	Syntax	Actual Code	Condition
beq	beq [src0] [src1] label	bnq [src0] [src1] Next j label Next:	Branching up or branching down more than 16 instructions
bne	bne [src0] [src1] label	beq [src0] [src1] Next j label Next:	Branching up or branching down more than 16 instructions
bgt	bgt [src0] [src1] label	blt [src0] [src1] Next j label Next:	Branching up or branching down more than 16 instructions
blt	blt [src0] [src1] label	bgt [src0] [src1] Next j label Next:	Branching up or branching down more than 16 instructions

3 Examples

3.1 Basic Use Examples

3.1.1 Loading an immediate into a register

```
cpy $t0 32      # Loads 32 into t0
```

3.1.2 Making a Procedure Call

```
rsh 8           # Switch to arguments schwap
cpy $h0 $t0     # Put argument0 in
cpy $h1 $s1     # Put argument1 in
# Store any wanted temporaries somewhere
jal Call
rsh 9           # Switch to return values schwap
cpy $s0 $h0     # Copy the return values out
```

3.1.3 Iteration and Conditionals

This is an example of which will iterate over 4 array elements in memory and add 32 to each of them. It will stop repeating after the 4 elements using beq.

```
        # There is a base memory address for an array in memory at s0
        cpy $t0 8
        cpy $t1 $z0
loop:
        r   $t2 0($s0)
        add $t2 32
        w   0($s0) $t2
        add $t1 2
        beq $t0 $t1 loop
```

3.2 relPrime and gcd Implementation

3.2.1 Assembly

```
RelPrime:
        rsh     8               #set schwap
        cpy     $s2 $ra         #save $ra
        cpy     $s0 $h0         #copy n out of schwap
        cpy     $s1 0x2         #load 2 to m
        rsh     8               #set schwap to args
While:
        cpy     $h0 $s0         #set a0 to n
        cpy     $h1 $s1         #set a1 to m
        jal     GCD             #call GCD
        rsh     9               #set schwap
        cpy     $t0 0x1         #load immediate 0x1 to t0
        bne     $h0 $t0 Done    #branch to done if r0 != 1
        add     $s1 0x1         #add 1 to m
        j       While           #jump to the start of the loop
Done:
```

```
        rsh     9               #load return registers
        cpy     $h0 $s1         #set r0 to m
        j       $s2 0           #return to the previous function
```

```
GCD:
        rsh     8               #schwap to argument register
Base:
        bne     $h0 $z0 GMain   #a!=0 go to GMain
        cpy     $t0 $h1         #copy h1 to t0 for RSH
        rsh     9               #schwap to return registers
        cpy     $h0 $t0         #load t0 to r1
        j       $ra 0           #return
GMain:
        beq     $h1 $z0 Exit    #jump to exit if b is zero
        bgt     $h0 $h1 If      #jump to If if a>b
Else:
        sub     $h1 $h0         #else: b=b-a
        j       GMain           #loop
If:
        sub     $h0 $h1         #if: a=a-b
        j       GMain           #loop
Exit:
        cpy     $t0 $h0         #copy h0 to t0 for rsh schwap
        rsh     9               #make sure we're in the right spot
        cpy     $h0 $t0         #copy t0 to h0
        j       $ra             #return
```

3.3 Machine Code

RelPrime		GCD	
PC	Hex	PC	Hex
00	3009	42	3009
02	063F	44	5C05
04	04CF	46	085F
06	150F	48	300A
08	0002	4A	0A8F
0A	3009	4C	2300
0C	0C4F	4E	4B05
0E	0D5F	50	1C0F
10	031F	52	0072
12	300F	54	2C00
14	1C0F	56	301F
16	0042	58	6CD7
18	2C00	5A	0DC2
1A	301F	5C	300F
1C	300A	5E	1C0F
1E	180F	60	004E
20	0001	62	2C00
22	4C85	64	301F
24	1C0F	66	0CD2
26	000A	68	300F
28	2C00	6A	1C0F
2A	301F	6C	004E
2C	1500	6E	2C00
2E	0001	70	301F
30	300F	72	08CF
34	1C0F	74	300A
36	000C	76	0C8F
38	2C00	78	2300
3A	301F		
3C	300A		
3E	0C5F		
40	2600		

4 Notes

4.1 Registers

4.1.1 Non-Schwappable

1. $z0 is reset on the rising edge of each CPU cycle, so it can be used for cycle-temporary storage.
2. The value in $pc should always be what the current instruction address is +2.

4.1.2 Schwappable

1. Possible uses for the reserved for future use groups:

Group Number	ID	Use
10	0	The constant 1
	1	The constant -1
	2	User set constant0
	3	User set constant1
11	0 - 3	I/O for devices 0 - 3
12	0 - 3	Syscall values 0 - 3
13	0 - 3	Kernel
14	0	Exception Cause
	1	Exception Status
	2	EPC
	3	Exception Temporary

4.2 Instructions

4.2.1 Types and Layouts

1. More of the types could be combined, but they will run faster if they are not.

4.3 Alias names

4.3.1 Registers

$z0: $0, $00, $zz, $zero

4.3.2 Instructions

orr: or

bne: bnq